

国标规定三相异步电动机在以下两种情况下不允许直接启动：

- ①额定容量大于7.5kW的；
- ②由专用变压器供电时，电动机容量大于变压器容量20%的；

不宜直接启动的电动机可以通过下列几种方法启动：

- ①采用适当功率的电机启动器；
- ②采用适当功率的变频器；
- ③采用星三角转换电路；

采用星三角转换电路启动是相对简单和节省成本的一种方式，它是通过三个接触器的配合，在电机启动初期使三相定子绕组构成星型连接（Y）。当电机运行一段时间后，再将三相绕组转换成三角形连接（ Δ ）。星型连接时加载在定子绕组上是相电压（220V），三角形连接时加载在定子绕组上的是线电压（380V），可见星三角启动是一种降压启动。

星三角降压启动适用于正常运行时定子绕组为三角形连接的电机（国标规定4kW以上的电机均应采用三角形接法），电路包括主回路和控制回路。控制回路可以由按钮、接触器、时间继电器完成，这种情况下不需要PLC参与。另外也可以通过PLC来实现定时/互锁等控制回路的功能，这种情况下控制回路的接线会相对简单一些，也可以节省一个时间继电器。无论是否使用PLC，主回路的接线保持不变。今天这篇文章，跟大家分享一个使用SCL语言编写的电机星三角启动的函数块。


SIEMENS
Ingenuity for life

西门子PLC高级编程语言SCL实例讲解

——电机星型-三角形启动

<https://www.founderchip.com>

本例程的电机主回路电路图如下图所示：



其中：

- Q1：断路器，有常开触点；
- KM1：主接触器，有常开触点；
- KM2：星型接触器，有常开触点；
- KM3：三角形接触器，有常开触点；
- FR1：热继电器，有常闭触点；

在博途环境下新建函数块FB5021_motorStartUp-star-delta，声明变量如下面两张图：

①输入/输出变量：

FB5021_motorStartUp-star-delta							
	名称	数据类型	默认值	保持	设定值	监控	注释
1	▼ Input						
2	start	Bool	false	非保持			启动按钮
3	stop	Bool	false	非保持			停止按钮
4	reset	Bool	false	非保持			复位按钮
5	fuse	Bool	false	非保持			断路器反馈信号（常闭，1=ok）
6	feedback_KM1	Bool	false	非保持			主接触器反馈信号（常开）
7	feedback_KM2	Bool	false	非保持			星型接触器反馈信号（常开）
8	feedback_KM3	Bool	false	非保持			三角形接触器反馈信号（常开）
9	feedback_FR1	Bool	false	非保持			热继电器反馈信号（常闭,1=ok）
10	timeChangeOver	Time	T#5s	非保持			星三角转换时间，默认5s
11	timeInternalFeedback	Time	T#1500ms	非保持			
12	▼ Output						
13	km1	Bool	false	非保持			主接触器线圈
14	km2	Bool	false	非保持			星型接触器线圈
15	km3	Bool	false	非保持			三角形接触器线圈
16	error	Bool	false	非保持			错误标志
17	▼ InOut						
18	<新增>						
19	▼ Static						

②静态变量：

19	▼ Static						
20	statStartRising	Bool	false	非保持			启动按钮上升沿信号
21	statStarRisingHF	Bool	false	非保持			启动按钮上升沿信号辅助变量
22	statStep	Word	16#0	非保持			控制状态
23	statEnabled	Bool	false	非保持			控制使能
24	statKM1	Bool	false	非保持			
25	statKM2	Bool	false	非保持			
26	statKM3	Bool	false	非保持			
27	statStep1 Activated	Bool	false	非保持			
28	statStartFeedback2	Bool	false	非保持			
29	statInternalFeedbackT...	Time	T#1500ms	非保持			
30	feedbackTON1	TON_TIME		非保持			
31	statFeedbackTimeON1	Bool	false	非保持			反馈1的时间到
32	changeOverTON	TON_TIME		非保持			
33	statTimeChangeOver...	Bool	false	非保持			
34	feedbackTON2	TON_TIME		非保持			
35	statStep4Activated	Bool	false	非保持			
36	statFeedbackTimeON2	Bool	false	非保持			
37	statStep7Activated	Bool	false	非保持			
38	feedbackTON3	TON_TIME		非保持			
39	statFeedbackTimeON3	Bool	false	非保持			
40	statError	Bool	false	非保持			
41	▼ Temp						

代码如下：

```

1  (*
2  Copyrights @Founderchip
3  =====
4  功能说明:
5      该程序实现电机的星型-三角形降压启动控制, 使用状态机实现逻辑控制。
6  输入参数:
7      start          : 电机启动按钮 (非自锁)
8      stop           : 电机停止按钮 (非自锁)
9      reset          : 复位按钮
10     fuse           : 电机断路器反馈 (1=ok)
11     feedback_KM1    : 主接触器 (KM1) 常开触点
12     feedback_KM2    : 星型接触器 (KM2) 常开触点
13     feedback_KM3    : 三角形接触器 (KM3) 常开触点
14     feedback_FR1    : 热继电器常闭触点 (1=OK)
15     timeChangeOver  : 星型-三角形切换时间, 默认5s
16     timeInternalFeedback : 内部判断反馈信号的时间, 默认1.5s
17  输出参数:
18     km1             : 主接触器线圈
19     km2             : 星型接触器线圈
20     km3             : 三角形接触器线圈
21     error           : 错误标志
22
23  作者:
24      北岛李工
25      2023-3-30
26  -----
27  修改日志:
28
29  2023-3-30 v1.0 版本(首发)          北岛李工
30
31  =====
32  *)

```

```

34  //启动按钮上升沿信号
35  #statStartRising := #start AND NOT #statStarRisingHF;
36  #statStarRisingHF := #start;
37  //使能
38  IF #fuse AND #feedback_FR1 THEN
39      #statEnabled := TRUE;
40  ELSE
41      #statEnabled := FALSE;
42      #statError := TRUE;
43  END_IF;
44  //复位
45  IF #reset THEN
46      #statError := FALSE;
47      #statStep := 0;
48  END_IF;
49  //停止
50  IF #stop THEN
51      #statStep := 10; //停止状态
52  END_IF;
53
54  IF #statEnabled THEN
55      //流程控制状态机
56  CASE #statStep OF //状态

```

```

57 CASE #statStep OF // 状态
58     0://步0—初始状态
59         //动作
60         #statInternalFeedbackTime := #timeInternalFeedback;
61         //转换
62         IF #statStartRising AND NOT #stop THEN
63             #statStep := 1;
64         END_IF;
65     1://步1-星型连接启动
66         //动作
67         #statKM1 := TRUE;
68         #statKM2 := TRUE;
69         #statStep1Activated := TRUE;
70         //转换
71         IF #statFeedbackTimeON1 THEN
72             #statStep := 2;
73         END_IF;
74     2://步2: 星型连接启动-判断反馈信号
75         //动作
76         //转换
77         IF NOT #feedback_KM1 OR NOT #feedback_KM2 THEN
78             #statStep := 3;
79         ELSE
80             IF #statTimeChangeOverON THEN
81                 #statStep := 4;
82             END_IF;
83         END_IF;
84     3://步3: 星型连接启动-反馈错误
85         //动作
86         #statKM1 := FALSE;
87         #statKM2 := FALSE;
88         #statStep1Activated := FALSE;
89         //转换
90         #statStep := 10; //转换到停止状态
91         #statError := TRUE;
92     4://步4: 星型连接转换到三角形连接
93         //动作
94         #statKM2 := FALSE;
95         #statStep4Activated := TRUE;
96         #statStep1Activated := FALSE;
97         //转换
98         IF #statFeedbackTimeON2 THEN
99             #statStep := 5;
100        END_IF;
101    5://步5: 星型—三角形转换-错误判断
102        //动作
103        #statStep4Activated := FALSE;
104        //转换
105        //判断星型接触器是否断开
106        IF #feedback_KM2 OR NOT #feedback_KM1 THEN
107            #statStep := 6; //星型接触器未断开或主接触器断开
108        ELSE
109            #statStep := 7; //星型接触器已断开
110        END_IF;
111    6://步6: 星型—三角形转换==>出现错误
112        //动作
113        #statKM1 := FALSE;
114        //转换
115        #statStep := 10; //转换到停止状态
116        #statError := TRUE;
117    7://步7: 星型—三角形转换==>正常启动三角形接触器
118        //动作

```

```

118         //动作
119         #statKM3 := TRUE;
120         #statStep7Activated := TRUE;
121         //转换
122         IF #statFeedbackTimeON3 THEN
123             #statStep := 8;
124         END_IF;
125     8://步8: 星型—三角形转换==>判断KM3的反馈
126         //动作
127         #statStep7Activated := FALSE;
128         //转换
129         IF NOT #feedback_KM3 OR NOT #feedback_KM1 THEN
130             #statStep := 9;
131         END_IF;
132     9://步9: 星型—三角形转换==>判断KM3的反馈
133         //动作
134         #statKM3 := FALSE;
135         //转换
136         #statStep := 10; //转换到停止状态
137         #statError := TRUE;
138     10://步10: 停止状态
139         //动作
140         #statKM1 := FALSE;
141         #statKM2 := FALSE;
142         #statKM3 := FALSE;
143         //转换
144         #statStep := 0;
145
146     END_CASE;
147 ELSE
148     #statKM1 := FALSE;
149     #statKM2 := FALSE;
150     #statKM3 := FALSE;
151     #statError := FALSE;
152 END_IF;

```

```

153 //定时器-判断KM1和KM2的反馈
154 □ #feedbackTON1(IN := #statStep1Activated,
155     PT := #statInternalFeedbackTime, //default time 1.5s
156     Q => #statFeedbackTimeON1);
157 //定时器-星型-三角形转换的时间
158 □ #changeOverTON(IN := #statStep1Activated,
159     PT := #timeChangeOver,
160     Q => #statTimeChangeOverON);
161 //定时器-
162 □ #feedbackTON2(IN := #statStep4Activated,
163     PT := #statInternalFeedbackTime,
164     Q => #statFeedbackTimeON2);
165
166 □ #feedbackTON3(IN:=#statStep7Activated,
167     PT:=#statInternalFeedbackTime,
168     Q=>#statFeedbackTimeON3);
169
170 //输出
171 #km1 := #statKM1;
172 #km2 := #statKM2;
173 #km3 := #statKM3;
174 #error := #statError;
175

```


这段代码我已经测试过了，如下图所示：

▼ 块标题： "Main Program Sweep (Cycle)"

注释

▼ 程序段 1： 测试

注释




如果你有任何问题，欢迎留言。

下面是西门子SCL编程的文章归档链接：

《《 西门子SCL编程入门到精通文章归档 》》

我的书《西门子S7-1200/1500 PLC SCL语言编程——从入门到精通》从硬件到软件，比较详细的介绍了SCL语言的编程，感兴趣的话可以扫描下面的二维码查看：



识别图中小
程序码购买